CBF program 161264  edda sveinsdottir

data:  n,/text/,gas,pCO2,hgb,tstart,remact,backgr,fmax,scalfac,
       dt,recirc,restime,infection,f(1),f(2),...,f(n).

_b_e_g_i_n _i_n_t_e_g_e_r n;  n:= in one;
_b_e_g_i_n _i_n_t_e_g_e_r i,j,h,g,K,C;  _b_o_o_l_e_a_n boole;
_r_e_a_l pCO2,hct,tstart,remact,backgr,fmax,scalfac,dt,recirc,restime,
  lg,lw,S,fi,w,v,sum,t,d,tstar,astar,bstar;
_a_r_r_a_y k,kk,a,x,VA[1:4],A[1:4,1:4],f[1:n];

_p_r_o_c_e_d_u_r_e gjr(a,n,eps)  exit:(singular);
_i_n_t_e_g_e_r n; _r_e_a_l eps; _a_r_r_a_y a; _l_a_b_e_l singular;
_b_e_g_i_n _i_n_t_e_g_e_r i,j,k; _r_e_a_l  pivot, z;
_a_r_r_a_y  b,c[1:n]; _i_n_t_e_g_e_r _a_r_r_a_y p,q[1:n];
_f_o_r k:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o
_b_e_g_i_n pivot:=0;
 _f_o_r i:=k _s_t_e_p 1 _u_n_t_i_l n _d_o
   _f_o_r j:=k _s_t_e_p 1 _u_n_t_i_l n _d_o
    _i_f abs(a[i,j]) > abs(pivot) _t_h_e_n
    _b_e_g_i_n pivot:= a[i,j]; p[k]:=i; q[k]:=j _e_n_d;
  _i_f abs(pivot) _< eps _t_h_e_n _g_o_ _t_o singular;
  _i_f p[k] |= k _t_h_e_n _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o
     _b_e_g_i_n z:= a[p[k],j]; a[p[k],j]:= a[k,j]; a[k,j]:=z _e_n_d j;
  _i_f q[k] |= k _t_h_e_n _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o
     _b_e_g_i_n z:= a[i,q[k]]; a[i,q[k]]:= a[i,k]; a[i,k]:=z _e_n_d i;
  _f_o_r  j:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o _b_e_g_i_n
    _i_f j=k _t_h_e_n _b_e_g_i_n b[j]:= 1/pivot; c[j]:=1 _e_n_d
           _e_l_s_e _b_e_g_i_n b[j]:= -a[k,j]/pivot; c[j]:= a[j,k] _e_n_d;
     a[k,j]:= a[j,k]:= 0 _e_n_d j;
  _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o
  _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o a[i,j]:= a[i,j] + c[i]×b[j]
_e_n_d k;
_f_o_r k:=n _s_t_e_p -1 _u_n_t_i_l 1 _d_o
_b_e_g_i_n
  _i_f p[k] |= k _t_h_e_n _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o
  _b_e_g_i_n z:= a[i,p[k]]; a[i,p[k]]:= a[i,k]; a[i,k]:=z _e_n_d i;
  _i_f q[k] |= k _t_h_e_n _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o
  _b_e_g_i_n z:= a[q[k],j]; a[q[k],j]:= a[k,j]; a[k,j]:=z _e_n_d j;
_e_n_d k;
_e_n_d gjr;

outcr;  outtext(|<<CBF-program 161264  e.s.|>);
outcr;  outcopy(|<<//|>);  outcr;
GAS:  K:= in char;  _i_f K|=34∧K|=162∧K|=23∧K|=151 _t_h_e_n _g_o_t_o GAS;
input(pCO2,hct,tstart,remact,backgr,fmax,scalfac,dt,recirc,restime);
INJ:  C:= in char;  _i_f C|=51∧C|=179∧C|=57∧C|=185∧C|=21∧C|=149 _t_h_e_n _g_o_t_o INJ;
hct:= hct/34;  _i_f K=34∨K=162
_t_h_e_n _b_e_g_i_n outcr;  outtext(|<<krypton 85|>);
  lg:= 0.92×0.843/(hct + 0.685×(1-hct));
  lw:= 1.26×0.843/(hct + 0.685×(1-hct)) _e_n_d
_e_l_s_e _b_e_g_i_n outcr;  outtext(|<<xenon 133_,|>);
  lg:= 0.77×0.744/(hct + 0.487×(1-hct));
  lw:= 1.44×0.744/(hct + 0.487×(1-hct)) _e_n_d;
S:= 1/(0.5×(1+exp((pCO2-40)/18.5)));
_i_f tstart=1 _t_h_e_n outtext(|<<_,_,_,digital output_,_,|>) _e_l_s_e _b_e_g_i_n
  outtext(|<<_,_,_,tstart:_,_,_,_,|>);  output(|<-n.dd|>, tstart) _e_n_d;
outtext(|<<_,_,_,st factor,S:|>);  output(|<-n.dd|>, S);
outtext(|<<_,_,_,remaining act:|>);  output(|<ndd.d|>,remact); outcr;
outtext(|<<aPCO2:_,_,|>);  output(|<dd|>, pCO2);  outtext(|<<_,_,_,fmax:_,_,_,_,|>);
output(|<ddddd.d|>, fmax);  outtext(|<<_,_,_,recirc corr:_,|>);
output(|<n.dd|>, recirc);  outtext(|<<_,_,_,background:_,_,_,|>);
output(|<ndd.d|>, backgr);  outcr;  outtext(|<<hgb:_,_,|>);
output(|<nd.d|>, hct×34);  outtext(|<<_,_,_,scal fact:|>);

output(|<dddd.d|>, scalfac);  outtext(|<<_,_,_,inj:_,|>);
outtext(_i_f C=51∨C=179 _t_h_e_n |<<car communis|> _e_l_s_e _i_f C=57∨C=185
  _t_h_e_n |<<_,car interna|> _e_l_s_e |<<_,vertebralis|>);
outtext(|<<_,_,_,resolv time:_,|>);  output(|<d.d_{10}-d|>, restime);  outcr;
outtext(|<<dt:_,_,_,|>);  output(|<n.dd|>, dt);

input(f);  _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o _b_e_g_i_n fi:= f[i];
  f[i]:= fi×scalfac×(1+fi×scalfac×restime)-backgr×scalfac _e_n_d;
fmax:= fmax×scalfac×(1+fmax×scalfac×restime) - backgr×scalfac;
remact:=remact×scalfac×(1+remact×scalfac×restime) - backgr×scalfac;
g:= 0;  i:= 0;
_f_o_r i:= i+1 _w_h_i_l_e ln(f[i+1]) > 0.5×(ln(f[i])+ln(f[i+2])) _d_o g:=i;
outtext(|<<_,_,_,data excluded:|>);  output(|<nd|>, g);
outtext(|<<_,_,_,data analysed:|>);  output(|<ndd|>, n-g);  outcr;  outtext(|<<

T W O - C O M P A R T M E N T  A N A L Y S I S
(a×exp(-alpha×t) + b×exp(-beta×t))|>);

_b_e_g_i_n _a_r_r_a_y y[1:n];  _r_e_a_l p,q;
_p_r_o_c_e_d_u_r_e Lsq(starti,interval);
_v_a_l_u_e starti,interval;  _i_n_t_e_g_e_r starti,interval;
_b_e_g_i_n _r_e_a_l X,XX,Y,XY,D;  X:=XX:=Y:=XY:= 0;
_f_o_r i:=starti _s_t_e_p 1 _u_n_t_i_l starti+interval _d_o _b_e_g_i_n
  X:= X+i×dt;  XX:= XX + (i×dt)|∧2;
  Y:= Y + y[i];  XY:= XY + i×dt×y[i] _e_n_d;
D:= (interval+1)×XX - X×X;
p:= ((interval+1)×XY - X×Y)/D;  q:= (XX×Y - X×XY)/D;
_e_n_d Lsq;

_f_o_r i:=entier(7/dt) _s_t_e_p 1 _u_n_t_i_l n _d_o y[i]:= ln(f[i]);
Lsq(entier(7/dt),n-entier(7/dt));  k[4]:= -p;  k[3]:= exp(q);  i:= g;
_i_f k[4]>0.23 _t_h_e_n _b_e_g_i_n outcr;  outcr;
outtext(|<<thalf of tail_,|>);  output(|<-nd.d|>, 0.693/k[4]);
outtext(|<<_,min_, too steep|>);  outcr;
outtext(|<<check log(tailpart-background) after 7 min|>);
_g_o_t_o END _e_n_d;
_f_o_r i:=i+1 _w_h_i_l_e i_<g+1+entier(5/dt) ∧ f[i]-k[3]×exp(-k[4]×i×dt)>0.1 _d_o
_b_e_g_i_n y[i]:= ln(f[i]-k[3]×exp(-k[4]×i×dt));  h:= i _e_n_d;
Lsq(g+1,h-(g+1));  k[2]:= -p;  k[1]:= exp(q);  outcr;  outcr;
outtext(|<<initial guess_,:_,|>);
output(|<-ndddd.dd|>, k[1]/fmax×100, k[2], k[3]/fmax×100);
output(|<-ndddd.ddd|>, k[4]);
_e_n_d of curvepeeling block;

_f_o_r i:=1,2,3,4 _d_o kk[i]:= k[i];  w:= 1;

IGEN:
_f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
_b_e_g_i_n VA[j]:= 0;  _f_o_r i:=j _s_t_e_p 1 _u_n_t_i_l 4 _d_o A[j,i]:= 0 _e_n_d j;
_f_o_r j:= g+1 _s_t_e_p 1 _u_n_t_i_l n _d_o
_b_e_g_i_n _i_f tstart=1 _t_h_e_n w:= 1/f[j];
  a[1]:= exp(-k[2]×j×dt);  a[2]:= -a[1]×k[1]×j×dt;
  a[3]:= exp(-k[4]×j×dt);  a[4]:= -a[3]×k[3]×j×dt;
  v:= f[j] - a[1]×k[1] - a[3]×k[3];
  _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o _b_e_g_i_n
    VA[i]:= VA[i] + w×v×a[i];
    _f_o_r h:=i _s_t_e_p 1 _u_n_t_i_l 4 _d_o A[i,h]:= A[i,h]+w×a[i]×a[h] _e_n_d i;
_e_n_d j;
_f_o_r i:=2,3,4 _d_o _f_o_r j:=i-1 _s_t_e_p -1 _u_n_t_i_l 1 _d_o A[i,j]:= A[j,i];
gjr(A,4,1_{10}-7,SINGULAR);  _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o _b_e_g_i_n sum:= 0;
_f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o sum:= sum+A[i,j]×VA[j];  x[i]:= sum _e_n_d i;
_f_o_r i:=1,2,3,4 _d_o k[i]:= k[i]+x[i];
_i_f k[1]>2×fmax ∨ k[1]<-0.01×fmax ∨ k[2]>5 ∨ k[2]<-0.1
  ∨k[3]>2×fmax ∨ k[3]<-0.01×fmax ∨ k[4]>1 ∨ k[4]<-0.1

_t_h_e_n _b_e_g_i_n outtext(|<<_,_,_,divergence|>);  boole:= _t_r_u_e;  _g_o_t_o FLOW _e_n_d;
_i_f abs(x[1]/kk[1])+abs(x[2])+abs(x[3]/kk[3])+abs(x[4]) > _{10}-4
_t_h_e_n _g_o_t_o IGEN;
outcr;  outcr;  outtext(|<<a,alpha,b,beta:_,|>);
output(|<-ndddd.dd|>, k[1]/fmax×100, k[2], k[3]/fmax×100);
output(|<-ndddd.ddd|>, k[4]);
boole:= _f_a_l_s_e;  t:= 0;  d:= 1;
TSTAR:  _i_f k[1]×exp(-k[2]×t) + k[3]×exp(-k[4]×t) > fmax _t_h_e_n
_b_e_g_i_n t:= t+d;  _g_o_t_o TSTAR _e_n_d;
t:= t-d;  d:= d/10;  _i_f d>0.001 _t_h_e_n _g_o_t_o TSTAR;  tstar:= t;
astar:=k[1]×exp(-k[2]×tstar);  bstar:=(k[3]-remact)×exp(-k[4]×tstar);

FLOW: _b_e_g_i_n _b_o_o_l_e_a_n digout;  _a_r_r_a_y r[1:n]; _i_n_t_e_g_e_r begi;
_r_e_a_l fg,fw,wg,ww,lambda,delaytime,begarea,area,ten,CBF;
tstar:= (g+2)×dt - (fmax-f[g+2])×dt/(f[g+1]-f[g+2]);
_i_f boole _t_h_e_n _b_e_g_i_n k[4]:= kk[4];  _g_o_t_o RAW DATA _e_n_d;
sum:= 0;  _f_o_r j:=g+1 _s_t_e_p 1 _u_n_t_i_l n _d_o _b_e_g_i_n
  r[j]:= k[1]×exp(-k[2]×j×dt) + k[3]×exp(-k[4]×j×dt) - f[j];
  _i_f tstart=1 _t_h_e_n w:= 1/f[j];  sum:= sum+r[j]|∧2×w _e_n_d j;
outcr;  outtext(|<<_,_,_,_,_,_,_,_,_,_,S.D.:_,|>);
output(|<-ndddd.dd|>, sqrt(abs(A[1,1])×sum/(n-4-g))×100/fmax, sqrt(
  abs(A[2,2])×sum/(n-4-g)), sqrt(abs(A[3,3])×sum/(n-4-g))×100/fmax);
output(|<-ndddd.ddd|>, sqrt(abs(A[4,4])×sum/(n-4-g)));
outcr;  outtext(|<<
calculated minus observed values in percent of fmax:|>);  outcr;
_f_o_r i:=g+1 _s_t_e_p 1 _u_n_t_i_l n _d_o _b_e_g_i_n
output(|<-ndd.d|>, r[i]/fmax×100);  _i_f i-g = (i-g)_:10×10 _t_h_e_n outcr _e_n_d;
fg:= k[2]×lg×100;  fw:= k[4]×lw×100;  k[4]:= kk[4];
wg:= (astar/fg)/(astar/fg + bstar/fw);  ww:= 1-wg;  outcr;  outcr;
outtext(|<<_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,×S|>);
outcr;  outtext(|<<rel weight grey_,:_,|>);  output(|<-n.dd|>, wg);
outtext(|<<_,_,_,_,_,flow grey_,:|>);  output(|<-ndddd.d|>, fg, fg×S);
outtext(|<<_,_,ml/100g/min|>);
outcr;  outtext(|<<rel weight white:_,|>);  output(|<-n.dd|>, ww);
outtext(|<<_,_,_,_,_,flow white:|>);  output(|<-ndddd.d|>, fw, fw×S);
outtext(|<<_,_,ml/100g/min|>);
outcr;  outtext(|<<weighted lambda_,:_,|>);  output(|<-n.dd|>, wg×lg+ww×lw);
outtext(|<<_,_,_,_,_,flow mean_,:|>);  output(|<-ndddd.d|>,
  wg×fg+ww×fw, (wg×fg+ww×fw)×S);  outtext(|<<_,_,ml/100g/min|>);
RAW DATA:  outcr;  outtext(|<<

N O N - C O M P A R T M E N T A L  A N A L Y S I S
(lambda for w-grey/w-white = 6/4 : |>);
lambda:= lg×0.6 + lw×0.4;  output(|<-n.dd|>, lambda);  outtext(|<<)|>);
digout:= tstart=1;  delaytime:= tstar - tstart;
_f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o f[i]:= f[i] - remact×exp(-k[4]×i×dt);
fmax:= fmax - remact×exp(-k[4]×0.5×(tstar+tstart));
outcr;  outcr;  outtext(|<<_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,_,×S
CBFr(_,10 min_,):|>);  recirc:= recirc×0.83;
begi:= _i_f digout _t_h_e_n 1 _e_l_s_e _i_f tstar<0 _t_h_e_n 1 _e_l_s_e entier(tstar/dt)+1;
ten:= _i_f digout _t_h_e_n 10/dt _e_l_s_e (10+tstart)/dt;
_i_f n<ten _t_h_e_n _b_e_g_i_n outtext(|<<_,_,_,not enough data|>);  _g_o_t_o next _e_n_d;
_i_f digout _t_h_e_n _b_e_g_i_n
area:= 0;  _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l ten _d_o area:= area + f[i];
area:= area×dt;  _g_o_t_o CBFten _e_n_d;
begarea:= area:= delaytime×fmax/dt+(begi-tstar/dt)×0.5×(fmax+f[begi]);
_f_o_r i:=begi _s_t_e_p 2 _u_n_t_i_l ten-2 _d_o
  area:= area+1/3×(f[i]+4×f[i+1]+f[i+2]);
area:= dt×(area + (ten-begi-entier(ten-begi)_:2×2)×f[ten]);
CBFten:  CBF:= (fmax-f[ten])×lambda×100/(area-recirc×fmax);
output(|<-ndddd.d|>,CBF, CBF×S);  outtext(|<<_,_,ml/100g/min|>);
next:  outcr;  outtext(|<<CBFr(infinity):|>);  recirc:= recirc×2;
_i_f digout _t_h_e_n _b_e_g_i_n
area:= 0;  _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n-2 _d_o area:= area+f[i];

area:= area×dt + (f[n-3]+f[n-2]+f[n-1]+f[n])/4/k[4]; _g_o_t_o CBFinf _e_n_d;
area:= begarea;  _f_o_r i:=begi _s_t_e_p 2 _u_n_t_i_l n-2 _d_o
area:= area + 1/3×(f[i]+4×f[i+1]+f[i+2]);
area:= area×dt + f[begi + (n-begi)_:2×2]/k[4];
CBFinf:  CBF:= fmax×lambda×100/(area-recirc×fmax);
output(|<-ndddd.d|>, CBF, CBF×S);  outtext(|<<_,_,ml/100g/min|>);
_e_n_d of FLOW block;  _g_o_t_o END;
SINGULAR:  outtext(|<<
normal matrix singular (the curve may be monoexponential)|>);
boole:= _t_r_u_e;  _g_o_t_o FLOW;
END:  outchar(11);
_e_n_d _e_n_d;

Ff
60
/Otto Lind f 050812 hø com ve temp inj 2 /
xenon
45.8
14.5
0.05
5.1
2.5
122
600
0.25
0.45
2_{10}-7
com

110,98,89,82,75,69,65,61,58,55,
52,49,47,45,44,43,41,39,38,37,
36,35,35,35,34,33,32,32,31,30,
30,29,29,29,28,28,28,26,26,26,
26,25,25,25,24,24,24,23,23,23,
22,22,22,22,21,21,21,21,21,21,

33
/Sonja Bjørkman f 310322 ve int ve temp, focal, inj 1 /
xenon
37.8
12.7
0.04
1.8
1.8
142
600
0.50
0.07
2_{10}-7
int

132,89,61,46,38,30,26,22,20,19,
18,16,16,15,14,12,10,10,9,9,
9,8,8,7,7,7,6,6,6,5,
5,4,3,